

```

;=====
; geoGopherScr: scrolling and paging routines
;=====
.if          Pass1
               .noeqin
               .include    geoGopherSym
               .include    geoGopherMac
               .include    geoGopher.inc
               .include    ultimate.inc
               .eqin

.endif
;=====
; Set up for gopher item scrolling. (Note: setCfg is in geoGopherCfg
; because it uses constants defined there).
;=====
setGoph:      LoadB      scrITop,15
               LoadB      scrIBot,186
               LoadW      scrILeft,312
               LoadB      thmbTop,23
               LoadW      thmbLeft,314
               LoadB      thmbMax,155
               LoadB      maxItems,7
               rts
;=====
; Set up for text (info item or file) scrolling.
;=====
setText:      LoadB      scrITop,7
               LoadB      scrIBot,194
               LoadW      scrILeft,308
               LoadB      thmbTop,15
               LoadW      thmbLeft,310
               LoadB      thmbMax,171
               LoadB      maxItems,1      ;count pages, not lines
               rts

```

```

=====
; Check for mouse click over scroll bar area.
=====
chkMouse:    lda        mouseData
             bpl        5$
             rts        ;ignore releases
5$           MoveB      thmbTop,r2L    ;click above scroll thumb?
             ldx        thumbSav
             dex
             stx        r2H
             MoveW      scrLLeft,r3
             lda        scrLLeft
             clc
             adc        #7
             sta        r4L
             lda        scrLLeft+1
             adc        #0
             sta        r4H
             php
             sei
             jsr        IsMseInRegion
             plp
             cmp        #$ff
             bne        10$
             jsr        doPgUp
             rts
10$          lda        thumbSav    ;click below scroll thumb?
             clc
             adc        thumbHi
             sta        r2L
             lda        scrLBot    ;r2H = scrLBot-7
             sec
             sbc        #8
             sta        r2H
             lda        scrLLeft    ;r3 = scrLLeft+1
             clc
             adc        #1
             sta        r3L
             lda        scrLLeft+1
             adc        #0
             sta        r3H
             lda        scrLLeft
             clc
             adc        #6
             sta        r4L
             lda        scrLLeft+1
             adc        #0
             sta        r4H

```

```

php
sei
jsr      IsMseInRegion
plp
cmp      #$ff
bne      20$
jsr      doPgDown
20$
rts
=====
;
doPgUp:   jmp      (pageUp)
doPgDown: jmp      (pageDown)
=====
; Draw (clear) scrollbar area.
=====
drawScrl: lda      #0          ;clear
          jsr      SetPattern
          lda      scrITop      ;r2L = scrITop+7
          clc
          adc      #7
          sta      r2L
          lda      scrIBot      ;r2H = scrIBot-7
          sec
          sbc      #7
          sta      r2H
          lda      scrLeft      ;r3 = scrLeft+1
          clc
          adc      #1
          sta      r3L
          lda      scrLeft+1
          adc      #0
          sta      r3H
          LoadW    r4,319
          jsr      Rectangle
          MoveB     scrITop,r3L
          MoveB     scrIBot,r3H
          MoveW     scrLeft,r4
          lda      #$ff        ;solid line
          jsr      VerticalLine
          rts
=====
; Draw (non-functional) top and bottom scroll arrows. For screen refreshes.
=====
doArrows: LoadW    r0,topArrow
          LoadB     r1L,39      ;X position in cards
          LoadB     r1H,14      ;Y position in pixels
          LoadB     r2L,1       ;width in cards
          LoadB     r2H,8       ;height in pixels
          jsr      BitmapUp
          LoadW     r0,botArrow
          LoadB     r1L,39      ;X position in cards
          LoadB     r1H,180     ;Y position in pixels
          LoadB     r2L,1       ;width in cards
          LoadB     r2H,8       ;height in pixels
          jsr      BitmapUp
          rts

```

```

; =====
; Dispatch routine for bottom button in scrollbar (gopher items).
; =====
itmBot:      lda      topltem      ;already at bottom?
            clc
            adc      maxItems
            cmp      numItems
            bcc      10$
            jsr      beep
            rts
10$          lda      numItems
            sec
            sbc      maxItems
            sta      topltem
            tax
            jsr      doltems
            rts

; =====
; Dispatch routine for top button in scrollbar (gopher items).
; =====
itmTop:      lda      topltem      ;already at top?
            bne      10$
            jsr      beep
            rts
10$          ldx      #0
            stx      topltem
            jsr      doltems
30$          rts

```

```

; =====
; Page down routine (called through otherPressVector).
; =====
itmPgDn:    lda        topltem
            clc
            adc        maxItems
            cmp        numItems    ;can page down at all?
            bcc        10$
            jsr        beep
            rts
10$         adc        maxItems
            sec
            sbc        #1
            cmp        numItems    ;can go down full page?
            bcc        20$
            lda        numItems    ;no, go to bottom
            sec
            sbc        maxItems
            sta        topltem
            bra        30$
20$         lda        topltem    ;yes, page down
            clc
            adc        maxItems
            sta        topltem
30$         ldx        topltem
            jsr        dolItems
            rts

; =====
; Page up routine (called through otherPressVector).
; =====
itmPgUp:    ldx        topltem
            bne        10$
            jsr        beep
            rts
10$         cpx        maxItems    ;can go up full page?
            bcc        20$
            lda        topltem    ;yes, page up
            sec
            sbc        maxItems
            sta        topltem
            bra        30$
20$         lda        #0
            sta        topltem    ;no, go to top
30$         ldx        topltem
            jsr        dolItems
            rts

```

```

;=====
; Draw thumb for scrollbar.
; destroyed: a8, a9 (from call to thumbPos)
;=====
doThumb:    jsr      thumbPos      ;sets thumb height and position
            sta      thumbSav      ;top of thumb
            lda      #0            ;clear
            jsr      SetPattern
            lda      scrITop        ;r2L = scrITop+7
            clc
            adc      #7
            sta      r2L
            lda      scrIBot        ;r2H = scrIBot-7
            sec
            sbc      #7
            sta      r2H
            lda      scrLeft        ;r3 = scrLeft+1
            clc
            adc      #1
            sta      r3L
            lda      scrLeft+1
            adc      #0
            sta      r3H
            lda      scrLeft        ;r4 = scrLeft+6
            clc
            adc      #6
            sta      r4L
            lda      scrLeft+1
            adc      #0
            sta      r4H
            jsr      Rectangle      ;clear scroll area
            lda      #1            ;solid
            jsr      SetPattern
            lda      thumbSav      ;already calculated
            sta      r2L
            clc
            adc      thumbHi
            sta      r2H
            MoveW    thmbLeft,r3
            ldx      #THUMB_WD      ;r4 = thmbLeft+(THUMB_WD-1)
            dex
            txa
            clc
            adc      thmbLeft
            sta      r4L
            lda      thmbLeft+1
            adc      #0
            sta      r4H
            jsr      Rectangle      ;draw thumb
            rts

```

```

; =====
; Calculate top position of thumb for scrollbar.
; Solve for offset from thmbTop:
; offset / (lowest thumb pos. - thmbTop) = topltem / max. top item
; (topltem * (lowest thumb pos. - thmbTop)) / max. top item = offset
;         pass:      numItems set
;         return:     sets thumbHi (height of thumb)
;                   returns top of thumb in .A
;         destroyed: a8, a9
; =====
thumbPos:  lda      maxItems      ;all items fit on screen?
           cmp      numItems
           bcc      10$
           MoveB    thmbMax,thumbHi
           lda      thmbTop
           rts
10$:       jsr      thumbSiz      ;sets thumbHi
           lda      scrBot
           sec
           sbc      #8           ;for arrow icon
           sbc      thumbHi      ;lowest possible thumb pos.
           sbc      thmbTop
           sta      a9L          ;src (byte)
           lda      topltem
           sta      a8L          ;dest (word)
           lda      #0
           sta      a8H
           ldx      #a8          ;dest
           ldy      #a9          ;src
           jsr      BMult        ;topltem * (lowest thumb pos. - thmbTop)
           lda      numItems
           sec
           sbc      maxItems
           sta      a9L          ;max. top item
           lda      #0
           sta      a9H
           ldx      #a8          ;dest
           ldy      #a9          ;src
           jsr      Ddiv         ;divide result by max. top item
           lda      thmbTop
           clc
           adc      a8L          ;and add to thmbTop
           rts

```

```

; =====
; Calculate height of thumb for scrollbar.
; Solve for thumb height:
; thumbHi / thmbMax = maxItems / numItems
; (maxItems * thmbMax) / numItems = thumbHi
; =====
thumbSiz:  lda      thmbMax
           sta      a9L
           lda      maxItems
           sta      a8L
           lda      #0
           sta      a8H
           ldx      #a8          ;dest
           ldy      #a9          ;src
           jsr      BBMult       ;maxItems * thmbMax
           lda      numItems
           sta      a9L
           lda      #0
           sta      a9H
           ldx      #a8          ;dest
           ldy      #a9          ;src
           jsr      Ddiv         ;(maxItems * thmbMax) / numItems
           lda      a8L
           cmp      #THUMBMIN
           bcs      10$
           lda      #THUMBMIN
           bne      30$
10$        lda      thmbMax
           cmp      a8L
           bcs      20$
           lda      thmbMax
           bne      30$
20$        lda      a8L
30$        sta      thumbHi
           rts

```